# The Simson Line

Let $ABC$ be a triangle. The lines $AB$, $BC$, $AC$ divide the plane into seven regions; let the point $P$ lie in the unbounded region containing the edge $AC$ in its boundary (see diagram). Let $X$, $Y$, $Z$ be the feet of the perpendiculars from $P$ to the lines $BC$, $AC$, $AB$ respectively.




**Simson's Theorem.** *The points $X$, $Y$, $Z$ are collinear if and only if $P$ lies on the circumcircle of $ABC$.*

*Proof.* The point $P$ lies on the circumcircle of $ABC$ if and only if

$$\angle APC = 180^o - \angle B. \tag{1}$$

Because the opposite angles at $X$ and $Z$ in the quadrangle $BXPZ$ are right angles we have

$$180^o - \angle B = \angle ZPX$$

so condition (1) is equivalent to

$$\angle APC = \angle ZPX \tag{2}$$

and on subtracting $\angle APX$ we see that (2) is equivalent to

$$\angle XPC = \angle ZPA. \tag{3}$$

A quadrilateral containing a pair of opposite right angles is *cyclic*, i.e. its vertices lie on a circle; in fact, the other two vertices are the endpoints of a diameter of this circle. Hence each of the quadrilaterals $AYPZ$, $BXPZ$, $CXPY$ is cyclic. Since the quadrangle $CXPY$ is cyclic we have

$$\angle XYC = \angle XPC. \tag{4}$$

Since the quadrangle $AYPZ$ is cyclic we have

$$\angle ZYA = \angle ZPA. \tag{5}$$

From (4) and (5) we conclude that (3) is equivalent to

$$\angle XYC = \angle ZYA. \tag{6}$$

But clearly (6) holds if and only if the points $X$, $Y$, $Z$ are collinear. $\square$

Here is a computer assisted coordinate calculation which proves Simson's Theorem. It shouldn't be to difficult to do by hand, especially if we take $\gamma = -\alpha$ below to simplify the formulas. We begin by loading the Maple package for doing linear algebra.

```
with(LinearAlgebra);
```

To calculate the foot $Z$ of the perpendicular from the point $P$ to the line $AB$ we use the formulas

$$Z = A + t(B - A), \qquad PZ \perp AB,$$

solve for $t$, and plug back in to get $Z$. Here is a Maple procedure to compute this.

```
foot:=proc(A,B,P) local t;
   t:=((B[1]-A[1])*(P[1]-A[1])+(B[2]-A[2])*(P[2]-A[2]))/
        ((B[1]-A[1])^2+(B[2]-A[2])^2);
   [A[1]+t*(B[1]-A[1]),A[2]+t*(B[2]-A[2])]
end proc;
```

We choose $A$, $B$, $C$, on the unit circle and $P$ arbitrarily.

```
A:=[cos(alpha),sin(alpha)];
B:=[cos(beta),sin(beta)];
C:=[cos(gamma),sin(gamma)];
P:=[x,y];
```

We use the procedure to calculate $X$, $Y$,$Z$:

```
X:= foot(B,C,P); Y:=foot(C,A,P); Z:=foot(A,B,P);
```

We define the matrix whose determinant vanishes when $X$, $Y$, $Z$ are collinear.

```
M:=Matrix([
                     [X[1], X[2], 1],
                     [Y[1], Y[2], 1],
                     [Z[1], Z[2] ,1]
                ]);
```

We compute its determinant.

```
W:=Determinant(M);
```

The determinant $W$ vanishes exactly when $X$, $Y$, $Z$ are collinear. The following commands show that $x^2 + y^2 - 1$ divides $W$ and that the quotient $m$ is independent of $P = (x, y)$.

```
m:=simplify(W/(x^2+y^2-1));
simplify(W-m*(x^2+y^2-1));
```

The last command evaluates to 0 and proves that $W = m(x^2 + y^2 - 1)$. Thus $X$, $Y$, $Z$ are collinear if and only if $x^2 + y^2 = 1$, i.e. if and only if $P$ lies on the circumcircle of $\triangle ABC$. The commands

```
mm:=expand(sin(alpha-beta)+sin(beta-gamma)+sin(gamma-alpha))/4;
simplify(m-mm);
```

produce an output of zero which shows that

$$m = \frac{\sin(\alpha-\beta)+\sin(\beta-\gamma)+\sin(\gamma-\alpha)}{4}$$

We have proved the following

**Algebraic form of Simson's Theorem.** *Let*

$$A = (\cos\alpha, \sin\alpha), \qquad B = (\cos\beta, \sin\beta), \qquad C = (\cos\gamma, \sin\gamma)$$

*be three points on the unit circle* $x^2 + y^2 = 1$*, let* $P = (x, y)$ *be an arbitrary point, and*

$$X = (x_1, x_2), \qquad Y = (y_1, y_2), \qquad Z = (z_1, z_2)$$

*be the feet of the perpendiculars from* $P$ *to the lines* $BC$*,* $CA$*,* $AB$ *respectively. Then*

$$\begin{vmatrix} x_1 & x_2 & 1 \\ y_1 & y_2 & 1 \\ z_1 & z_2 & 1 \end{vmatrix} = m(x^2 + y^2 - 1)$$

*where*

$$m = \frac{\sin(\alpha-\beta)+\sin(\beta-\gamma)+\sin(\gamma-\alpha)}{4}.$$